
Documentation Conventions
This guide uses the following documentation conventions.

Text and Programming Code

Example Description Explanation

ready()
grip_close()
finish()

evenly
spaced
computer
font

Programming code. In syntax
sections, required characters that
must be included.

gripdist_set(distance)
motor(axis,pulses,c)
if expression

italics User supplied item. Can be simple
(integer, variable) or complex
(expression, statements)

align_X|align_Y
M_READ|M_WRITE
X|Y|Z

vertical pipe
or bar

A choice between two or more
items. One must be chosen unless
it is optional (in square brackets).

place[3]
message[2,2]
data[10,4,7]

square
brackets
in arrays

Required characters of array
syntax. Must be included.

grip_close([force])
home([axis][,axis])
...[flags] [x|X]...

square
brackets
in any other
part of code

Optional items in code. Can be
included or omitted depending on
the needs of the program.

lock(7)
...
unlock(7)

three dots
on one line
or
on three
lines

Omitted code of the example. A
place for additional material which
is not specified.

\ (backslash)
_ (underscore)
" (double quote)

character(s)
with
description(s)
in
parentheses.

Characters referred to in the text
which need to be clearly identified.

use with
to end
when here

bold Names of commands, functions,
keywords, etc. used in the text
which could be confused.



RAPL-3 Reference Guide  3

Commands and Keywords
The following documentation conventions are used for

� all subroutines, functions, and commands in libraries

� all flow control statements

� other keywords (main, return, comment, sizeof)
 

 name_of_command/keyword
 Description  A description of the functionality of this subroutine, function, command, control

statement, or keyword.

 Details of usage.

 Caution  Any characteristics that could create a problem.

 Syntax  Required characters are in non-italic monospace font. Programmer-
supplied identifiers and constructs are in italics. Optional items are in [square
brackets]. Long lines may carry over onto a second line on the printed page, but
in a program must be written either on one line or with a \ (backslash) line
continuation character.

 Subroutines, functions, and commands are given in declaration form.

 Parameters
Arguments

 A list with explanations and types.

 Where a parameter is a standard-library defined enum or struct, the members
are listed.

 Returns  The return value of the function or command which also indicates success or
error.

 Example  An example of use in a program.

 Result  The example’s result, if applicable.

 See Also  Any related RAPL-3 commands, functions, subroutines, statements, keywords, or
topics, described in this Reference Guide.

 System Shell
Application Shell

 An equivalent command in the CROS/RAPL-3 system shell or application shell,
described in the Application Development Guide.

 RAPL-II  Any similar RAPL-II commands.

 Category  The category of this and related commands which are listed in the category
section.

 Related Resources
 Related material can be found in these documents.

� Release notes on the diskettes.

� Application Development Guide
A guide for developing your robotic application using all components of your
robot system: arm, controller, teach pendant, personal computer, Robcomm3,
RAPL-3 programs, application shell, and system shell.

� F3 Robot System Installation Guide

� A465 Arm and C500 Controller User Guides

� A255 Arm and C500 Controller User Guides

 General Program Format

All RAPL-3 programs follow the same general format. Some elements are
required. Other elements are optional depending on the complexity of the
program.

Example 1: Basic Program in RAPL-II Style
 A basic program can contain

� only a main function

 and follow a style similar to RAPL-II

� implicit declarations of variables

� familiar RAPL-II command names

 main function  main ;; begin program

fast = 50 ;; implicitly declare and initialize integers
slow = 25
z = 1

speed(fast) ;; set speed
move(_safe) ;; move and implicitly declare cartesian location

do ;; begin do loop

appro(_a,5) ;; pick from location a, implicitly declare location
grip_open(100)
grip_finish()
move(_a)
finish()
grip_close(100)
grip_finish()
depart(5)

move(_safe) ;; move to safe location between pick and place

appro(_b,5) ;; place at location b, implicitly declare location
move(_b)
finish()
grip_open(100)
grip_finish()
depart(5)

move(_safe) ;; move to safe location between place and pick
z = z + 1 ;; increment counter in loop

until z == 10 ;; condition to end do loop

end main ;; end program

 Example 2: Basic Program in Preferred RAPL-3 Style
 A basic program can contain

� a main function

� a subroutine

 and follow the preferred style of RAPL-3

� explicit declarations of variables, including teachables

 subroutine

main function

 sub io(int out_channel, int out_state, int in_channel)
int in_state
output(out_channel, out_state)
do

delay(250)
input(in_channel, in_state)

until (in_state) == 1
end sub

main

int i ;; explicitly declare variables
teachable int fast, slow, cycles ;; explicitly declare teachable variables
teachable cloc safe, a, b ;; explicitly declare teachable locations

move(safe)
speed(fast)

for i = 1 to cycles ;; use a for loop
;; cycles is teachable, set outside

appro(a,5)
grip_open(100)
io(1,1,2)
speed(slow)
move(a)
grip_close(100)
depart(5)

speed(fast)
move(safe)

appro(b,5)
io(3,1,4)
speed(slow)
move(b)
grip_open(100)
depart(5)

speed(fast)
move(safe)

end for

end main



8 General Program Format

 The Main Program
 Every RAPL-3 program contains a main function.
 

 main
 Description  A required function for each program. Requires main and end main to indicate

the beginning and the end of the main function.

 main is the place in the program where execution begins.

 The main function may not call itself.

 Syntax  main
statement(s)

end main

 Returns  Main does not have to explicitly return a value. By default, 0 (zero) is returned.
Any integer could be returned.

 Example  main
teachable cloc pick, place
move(pick)
grip_close()
move(place)
grip_open()

end main

 RAPL-II  RAPL-II did not have a function or structure similar to main. RAPL-II’s STOP
command had a purpose similar to end main.

 Lines of a Program
 A RAPL-3 program consists of a number of lines of ASCII text. Statements and
declarations are terminated by the line end.

 Line Continuation

 To continue on the next line, end a line with the \ (backslash) character. For
example

 a = b + c + d \
+ e + f

 is read as one statement.

 Without the continuation character

 a = b + c + d
+ e + f

 the first part of the statement ends at the end of the first line and is read as a
statement. The second part is a fragment which causes a syntax error when
compiling.

 Lines that end with , (a comma) are automatically considered to be continued.
For example,

 printf(�The coordinates are {}, {}, {}\n�,
x, y, z)

 Comments
 A comment starts with ;; (two semicolons) and extends to the end of the line. A
comment can start at the beginning of a line or after some program code. For
example:

 ;; calculate the position error:

 x_error = x_pos - desired_x_pos ;; for the x-axis
y_error = y_pos - desired_y_pos ;; for the y-axis
z_error = z_pos - desired_z_pos ;; for the z axis

Labels
 A statement can be marked with a special identifier called a label. The label has ::
(two colons) after the identifier. A labels is used as the target of a goto statement.

 Syntax
 label_identifier:: statement

 where
label_identifier  is the name of the label and follows the rules for identifiers,

and
statement  is the statement line being labelled.

 The statement can be an empty line.

 Examples
 my_label:: current_location = num

start_again::

 Keywords
 The following identifiers are keywords of RAPL-3. They are reserved for the RAPL-
3 language and cannot be redefined.  In particular, the following keywords
cannot be used as the name of any variable, subroutine, function, or command:

and

break

_builtin

case

cloc

command

comment

const

continue

do

else

elseif

end

enum

except

export

float

for

func

global

gloc

goto

if

ignore

import

int

libversion

loop

main

mod

not

of

or

ploc

private

proto

raise

resume

return

retry

sizeof

static

step

string

struct

sub

teachable

then

to

try

typedef

union

unteachable

until

var

void

volatile

while

with

 Data Types and Variables

 RAPL-3 programs can work with many different types of data and also permits
user-defined data types. This chapter presents the basic data types supported by
RAPL-3, and goes on to look at the kinds of user-defined types that can be
constructed.

 Basic Data Types
 RAPL-3 supports the following basic data types.

 Name  Description  Size (bytes)

 int  32-bit signed integer
(Range: -2147483648 to +2147483647)

 4

 float  IEEE single precision floating point
(Range: –1.7 x 10–38)

 4

 string  variable length string
(Range: 0 to 65535  8-bit characters)

 4 + number of
characters

 cloc  cartesian location  36

 ploc  precision location  36

 void  used for forming generic pointers  �

 

 int
 An int, or integer, is a signed number without any decimal or fractional part.
Examples: 0, 1, 23, 456, -7, -89

 float
 A float, or floating point number, is a number with a decimal or fractional part
and an optional exponent.  A float has up to seven significant digits.
Examples: 4.75, -99.99, 1.0, 3.141593, 1.0e10

 string
 A string is a set of characters:  uppercase or lowercase letters, digits,
punctuation and other graphic characters, and the blank space.  In a string, a
digit is a character and does not have numeric value as it does in a number (int
or float). RAPL-3 does not have a character data type.

 cloc
 A cloc, or cartesian location, represents a point in the robot arm workspace
defined by cartesian co-ordinates. Coordinates have three translational elements
(along axes) x, y, and z, and three rotational elements (around axes) z, y, and x.
The values of a cloc are independent of arm position and arm type.

 ploc
 A ploc, or precision location, represents a point in the robot arm workspace
defined by increments of rotational movement, specifically encoder counts, of
each joint of the arm and any additional axes (j1, j2, j3, j4, j5, j6, j7, j8). The
values of a ploc are dependent on the robot.

 void
 The void type is used to form void pointers (pointers that can point to any type).

void@ x
Void pointers are assignment compatible with all other types of pointers.

 Identifiers
 An identifier is used for the name of a variable, type, subroutine, function, or
command.

 Character Set
 An identifier begins with a letter. This may be followed by zero or more letters,
digits, or _ (underscore) characters.

 a b c d e f g h i j k l m n o p q r s t u v w x y z
A B C D E F G H I J K L M N O P Q R S T U V W X Y Z
0 1 2 3 4 5 6 7 8 9
_

 Case
 Letters may be either uppercase (ABCDE), lowercase (abcde), or mixed (AbCdE).
RAPL-3 is case-sensitive with identifiers. For example, the following are all
different identifiers.

 x ;; lowercase
X ;; uppercase
symbol ;; lowercase
SYMBOL ;; uppercase
sYmBoL ;; mixed
SyMbOl ;; mixed

 Length
 An identifier may be any length, but only the first 32 characters are significant.
For example, the following are not different identifiers.

 location_sensor_data_collection_1
location_sensor_data_collection_2

 Examples
 There are many possibilities of valid identifiers.

 Valid
 a ;; a single letter
num ;; several letters
my_symbol ;; letters with underscore
MySymbol ;; letters of different cases
x3 ;; letter with digit
rack_loc_12 ;; letters, underscores, digits

 Invalid
 3a ;; begins with a digit, not a letter
my$symbol ;; uses a character not in the valid character set
&num ;; uses a character not in the valid character set

;; and does not begin with a letter

 Declarations
 This section details the declaration of: int, float, string, cloc, and ploc. For the
declaration of arrays of these types, see the Arrays section. For const, see the
Initializers section.

 Each variable must be declared as one specific type of variable (int, float, string,
cloc, ploc, const). A declaration states the type of variable and the name of the
variable.

 You can declare a variable explicitly or implicitly. It is good programming practice
to explicitly declare all variables.

 Explicit Declarations
 When you declare a variable explicitly, you list it in a declaration statement
before you use it in the program.

 Variables being declared as the same type can be listed in the same declaration,
separated by commas.

 Syntax

 type identifier
type identifier, identifier, identifier ...

 where
type is the data type, and
identifier is the name of the variable and follows the rules for identifiers.

 Examples

 Type  Example  Description

 int  int i  i is an integer

 float  float a,b  a and b are floats

 string  string[10] message  message is a string that can hold 10 or fewer
characters

 cloc  cloc pick_1,
place_1

 pick_1 and place_1 are cartesian locations

 ploc  ploc pick_2,
place_2

 pick_2 and place_2 are precision locations

 

 Implicit Declarations
 When you declare a variable implicitly, you indicate the variable’s type with a
prefix before its name when you use it in the program for the first time.

 If a variable is used without having been explicitly declared, the compiler looks
for an implicit declaration prefix character on the variable name to determine the
type of variable. If there is no prefix character, the compiler defines the variable
as the default type, an int, and issues a warning.

 In general, implicit declarations should be avoided. You should always explicitly
declare variables.

 Syntax

 [prefix_character]identifier

 where
prefix_character is the character indicating the data type, and
identifier is the name of the variable and follows the rules for identifiers.

 Implicit Declaration Prefix Characters

 Prefix Character  Type  Example

  none  int  a = 2

 %  percent sign  float  %b = 10.25

 $  dollar sign  string[64]  $m = "Robot working.\n"

 _  underscore  cloc  here _z

 #  number sign  ploc  here #y

 Examples

 Type  Example  Description

 int  e = c + d  e is defined as an int, if it has not been seen before.

 float  %h = f * g  h is defined as a float.

 string  $notice9 =
"stop"

 notice9 is defined as a string[64].

 cloc  here(_place22)  place22 is defined as a cloc.

 ploc  here(#material3
3)

 material33 is defined as a ploc.

 Implicit with Explicit
 If an implicit declaration prefix is used in an explicit declaration statement, the
implicit prefix is ignored by the compiler. For example,

 float %b ;; the variable b is declared as a float
float $c ;; the variable c is declared as a float
float #d ;; the variable d is declared as a float

 Identifiers
 The prefix character indicates the type of declaration. It is not part of the
identifier, the variable’s name. For example, if _m was used in a statement, a cloc
with the name m was defined. A later statement with #m causes an error, the
same way that cloc m followed by ploc m causes an error.

 Scope

 Two variables with the same scope cannot have the same name. For definitions of
scope, see the Scope section of the Subprogram chapter.

 Teachables

 Teachable variables that are declared inside a sub, func, or command must not
have the same name as any teachable outer-frame variable.

 Strings
 The string type is essentially a character array with a fixed size.

 The string type must always have a subscript, indicated by [ ] (square brackets).

 String[number]
 Usually, the subscript contains a number to specify the maximum length of
string that can be stored in it, such as string[10] or string[64].

 Syntax
 string[number] identifier

 where
string and the square brackets are required,
number is the character size of the string, and
identifier is the name of the variable and follows the rules for identifiers.

 String[ ]
 In some circumstances, the subscript can be empty.

 string[]

 This undimensioned string declaration can be used only in the following
circumstances.

� A simple single string being initialized. When string[] is used, the compiler
determines the size of the string. In this example, the compiler makes notice9
a string[18].

 string[] notice9 = �End of work cycle.�

� A function formal parameter or var parameter.

 func int strlen(string[])
sub str_append(var string[] dst, string[] src)

� The target of a pointer.

 string[]@ sptr

 For a table of pointers to strings of unknown length, use

 string[]@[5] greek = {�alpha�, �beta�, �gamma�, �delta�,
�epsilon�}

 

Notes:
A RAPL-3 string is actually stored as a length, a limit, and an array of

characters.  The length value indicates how many characters are actually valid.
Strings can be created with at most space for 65,532 characters.  The limit value
indicates how many characters there is actually room for.  For example, if we
have a variable:

string[10] s
then s is initially created with its length set to 0 (no characters; the empty string)
and its limit set to 12.  The limit is 12 because RAPL-3 always allocates storage in
units of 1 word (or 4 characters); string[10] actually needs 1 word for the length
and limit, and an additional 3 words for the characters (which actually is 3 * 4 or
12 characters in size.)  After this statement:

s = �hello!�

the length of s is set to 6, and the characters �h�, �e�, �l�, �l�, �o� and �!� have been
stored in the character part of the string.

 Termination
 RAPL-3 does not use any string termination character. The variable is declared
and the string of characters is packed into the variable.

 Concatenation
 To concatenate (link together to form a longer string), use the str_append
subroutine with string variables. The + (plus) operator can be used to
concatenate string constants.

 Arrays
 An array is a collection of data objects where all are the same data type and all
use the same identifier but each has a unique subscript.

 Syntax
 base_type[ subscript_list ] identifier

 where
base_type is the data type of each element in the array,
subscript_list is a comma-separated list of one or more constant expressions

defining each dimension, and
identifier is the name of the variable and follows the rules for identifiers.

 A subscript must be a constant expression, such as a simple integer constant.
The compiler must be able to compute the value of each constant expression at
compile time.

 Types

 You can have an array of any type or an arrays of arrays.

 Dimensions
 There is no limit on the number of dimensions allowed, except for teachable
arrays. See Teachables.

 Numbering

 In RAPL-3, numbering begins with 0.

 Declaration  Number of
Elements

 Numbering

 int[4] a  4  a[0],   a[1],   a[2],   a[3]

 int[10] a  10  a[0],   a[1],   a[2],   a[3],   a[4],   a[5],   a[6],   a[7],   a[8],   a[9]

 int[20] a  20  a[0],   a[1],   a[2],  a[3],   a[4],   a[5],   a[6],   a[7],   a[8],   a[9],
a[10], a[11], a[12], a[13], a[14], a[15], a[16], a[17], a[18], a[19]

 int[100]
a

 100  a[0],   a[1],   a[2],  a[3],   a[4],   a[5],   a[6],   a[7],   a[8],   a[9],
through to
a[90], a[91], a[92], a[93], a[94], a[95], a[96], a[97], a[98], a[99]

 Review of Strings

 Example  Description

 string[30] z  a string that can hold 30 or fewer characters

 One Dimensional Arrays

 Example  Description

 int[5] a  an array of 5 integers
  a[0], a[1], a[2], a[3], a[4]

 float[10] b  an array of 10 floats
  b[0], b[1], b[2], ... b[9]

 ploc[20] c  an array of 20 precision locations
  c[0], c[1], c[2], ... c[19]

 string[30] [10] d  an array of 10 strings
  d[0], d[1], d[2], ... d[9]
  each can hold 30 or fewer characters

 Two Dimensional Arrays

 Example  Description

 int[5,10] e  a 2-dimensional array of 50 integers
  e[0,0] ... e[0,9]
  ...           ...
  e[4,0] ... e[4,9]

 float[10,20] f  a 2-dimensional array of 200 floats
  f[0,0] ... f[0,19]
  ...           ...
  f[9,0] ... f[9,19]

 ploc[5,10] g  a 2-dimensional array of 50 precision locations
  g[0,0] ... g[0,9]
  ...           ...
  g[4,0] ... g[4,9]

 string[20][5,10]
h

  a 2-dimensional array of 50 strings
  h[0,0] ... h[0,9]
  ...           ...
  h[4,0] ... h[4,9]
  each can hold 20 or fewer characters

 int[10] [5] i  a 2-dimensional array of 50 integers
  same as int[5,10] e
  brackets are applied from left to right

 float[20][10] j  a 2-dimensional array of 200 floats
  same as float[10,20] f
  brackets are applied from left to right

 string[20] [10]
[5] k

 a 2-dimensional array of 50 strings
  same as string[20] [5,10] h

 string[50][23 +
7] m

 an array of 30 strings,
  each can hold 50 or fewer characters

 

 Multi Dimensional Arrays

 Example  Description

 int[2,2,2] n  a 3-dimensional array of integers

   n[0,0,0], n[0,0,1],
  n[0,1,0], n[0,1,1],

   n[1,0,0], n[1,0,1],
  n[1,1,0], n[1,1,1]

 float[5,5,5,5] p  a 4-dimensional array of integers
  p[0,0,0,0] to p[4,4,4,4]

 

 Declarations

 You cannot implicitly declare an array.

 However, if you use the implicit declaration syntax in a statement with an array,
you will not cause a problem, if the array is previously declared and the implicit
declaration character matches the base type of the array. For example,

 ploc[16,16] a
...
here(#a[1,1])

Teachables
 A variable that is teachable is accessible from outside the program.

 Use
 Teachables provide an easy way, outside the program, to modify a value for a
variable, store that value, and use the value in a program. Using this feature
avoids writing (hard-coding) values in the program and having to re-write the
program to change the values. It also avoids storing the values in a custom user-
designed file and having to carefully edit the file to change values and include a
routine in the program to read that custom data file.

 Data about teachable variables and their values are stored in the variable file.
When you run a program, the operating system takes the program�s variable file
and uses its values to initialize the variables in the program just before running.

 Variable (v3) File
 Data about teachable variables are stored in the variable file (also known as a v3
file). You modify data, or �teach� locations and other variables, using the teach
pendant or the application shell.

You can create a variable file in a number of ways.

� Refreshing from the Program.  When your program file is in a CROS directory
(in CROS-500 or CROSnt), ash�s refresh command reviews the program and
adds any teachable variables of the program to ash�s database. After
assigning values (including teaching locations) to the teachables in the
database, this new data is saved to the variable file. This method is used if
you write your program before teaching your locations.

� Building Independently.  You can build a variable file completely in a CROS
directory (in CROS-500 or CROSnt) using ash or the teach pendant. With
ash�s or the teach pendant�s database, you create variables and assign values
to them. When you are finished this data is saved to in the variable file. This
method is used if you teach your locations before writing your program.

See the Application Development Guide chapters on the application shell.

 Declarations
 You make a variable teachable by adding the keyword �teachable� before the data
type at declaration. Teachables are not initialized.

 Syntax

 teachable  type identifier
teachable  type identifier, identifier, identifier ...

 where
teachable is a necessary keyword
type is the data type, and
identifier is the name of the variable and follows the rules for identifiers.

 Examples

 Example  Description

 teachable int cycles  cycles is an teachable integer

 teachable float a, b, c  a, b, and c are teachable floats

 teachable string[10] note  note is a teachable string that can hold 10 or fewer
characters

 teachable cloc pick_1,
place_1

 pick_1 and place_1 are teachable cartesian
locations

 teachable ploc pick_2,
place_2

 pick_2 and place_2 are teachable precision
locations

 teachable int[3] step  step is a teachable array of 3 integers: step[0],
step[1], step[2]

 teachable float[5,5]
delta

 delta is a teachable two-dimensional array of
floats: delta[0,0] ... delta[4,4]

 teachable ploc[2,10] spot  spot is a teachable two-dimensional array of
precision locations: spot[0,0] . . . spot[1,9]

 Limitations
 Data Types

 There are limits on which data types are teachable. Simple, scalar variables can
be teachable. One-dimensional arrays of variables can be teachable. Two-
dimensional arrays, except string[n], can be teachable. Three-dimensional and
higher dimensional arrays cannot be teachable. The void type cannot be
teachable.

  = can be teachable

�  = cannot be teachable

  int  float  string[n]  cloc  ploc  gloc  void

 simple        �

 one-dimensional
array

       �

 two-dimensional
array

   �     �

 three-dimensional
or higher array

 �  �  �  �  �  �  �

 

 Not Initialized

 A variable cannot be both teachable and initialized. You cannot write

 teachable int a = 5
teachable string[64] message_12 = �Error recovery underway.�.

 Storage Class: Static

 Variables which are declared as teachable are static. They should not be used in
recursive routines except as read only.

 Defaults and Unteachables
 Scope and Declaration Defaults

 The following variables are teachable by default.

 Local (within a subprogram or main) and Implicitly Declared

� clocs, and plocs

sub
...

here(_point)
end sub

main
...

here(_place)
end main

 Outer-Frame (outside all subprograms and main) and Explicitly Declared

� clocs, and plocs

� 1-dimensional and 2-dimensional arrays of clocs, and plocs

ploc start_point
cloc[10] point

sub
...
end sub

main
...
end main

All other variable types are unteachable by default.

Unteachable Declaration

A variable can be declared as unteachable with the unteachable keyword. This
can be used to make an outer frame location that is not teachable, for example

unteachable cloc[10] point

sub
...
end sub

main
...
end main

Location Constants
You can initialize cloc and ploc variables.  The RAPL-3 compiler has built-in
functions: cloc{} for generating cloc constants, and ploc{} for generating ploc
constants.  All of the arguments to cloc{} or ploc{} must be constant expressions
and the result is a constant expression that can be used in a variable
initialization.

The format of cloc{}  is:

cloc{ flags, x, y, z, zrot, yrot, xrot, e1, e2 }

Where flags specifies extra information about the location, x, y, and z are the
translational coordinates along the world axes, zrot, yrot, and xrot are
orientational coordinates around world axes, and e1 and e2 are the coordinates
for extra axes such as track.  The argument flags must be an int constant
expression and all other arguments are float constants.

An example of cloc{} is the following definition:

cloc my_tool = cloc{0, 0.0, 0.0, 10.0, 0.0, 0.0, 0.0, 0.0, 0.0}
;; tool transform for use with the tool_set() command.

The format of ploc{}  is:

ploc{ machtype, flags, a1, a2, a3, a4, a5, a6, a7, a8 }

Where machtype is the machine type (each type of machine, F3, A465, A255, ...,
has a different geometry and configuration resulting in different encoder counts
for a given location), flags specifies extra information about the location, and a1
to a8 specify the number of encoder pulses from zero of the desired position for
each axis.  The arguments machtype, flags, and a1 to a8 are integer constant
expressions.

An example of ploc{} is:

ploc start_point = ploc{mc_a465, 0, 3500, 2800, 1000, 4000,
2500, 1500}

;; initialized precision location

A word of warning: initialized clocs are useful for specifying tool transforms and
related information.   It is, however, very dangerous to hand-construct plocs and
command to robot to move to them.  This is because the robot cannot physically
move to any arbitrary joint configuration, and may collide with itself or objects in
the workspace.  If you must hand-construct locations, use extreme care.

Control Flow

When a pr ogram execut es, gener ally t he comput er execut es one line,  t hen t he
next , the n the  next .  In o rder to  m ake a program do u sefu l things  — for example,
to repeat  a partic u lar  ta sk  10 tim es — we m ust  be able to  alter the  way in whic h
cont rol passes f rom line t o line of  t he pr ogram.

This sect ion deals w ith st atement s t hat  alt er  t he sequence in w hich t he
statement s in a pr ogram execut e, allow ing loops and condit ional st atement s.

